Problem:
Score : 500 points

Problem Statement

Seisu-ya, a store specializing in non-negative integers, sells N non-negative integers. The i-th integer is A_i and has a utility of B_i.
There may be multiple equal integers with different utilities.

Takahashi will buy some integers in this store. He can buy a combination of integers whose bitwise OR is less than or equal to K. He wants the sum of utilities of purchased integers to be as large as possible.

Find the maximum possible sum of utilities of purchased integers.

Constraints

1 \leq N \leq 10^5

0 \leq K < 2^{30}

0 \leq A_i < 2^{30}(1\leq i\leq N)

1 \leq B_i \leq 10^9(1\leq i\leq N)

All input values are integers.

Inputs

Input is given from Standard Input in the following format:

N K
A_1 B_1
:
A_N B_N

Outputs

Print the maximum possible sum of utilities of purchased integers.

Sample Input 1

3 5
3 3
4 4
2 5

Sample Output 1

8

Buy 2 and 3 to achieve the maximum possible total utility, 8.

Sample Input 2

3 6
3 3
4 4
2 5

Sample Output 2

9

Buy 2 and 4 to achieve the maximum possible total utility, 9.

Sample Input 3

7 14
10 5
7 4
11 4
9 8
3 6
6 2
8 9

Sample Output 3

32

Solution in OCaml (4.02.3) (judge verdict: Accepted)

let flood_bits bits =
  List.fold_left (fun bits shamt ->
    bits lor (bits lsr shamt)) bits
  [1; 2; 4; 8; 16; 32]
let highest_bit bits =
  bits land lnot (flood_bits bits lsr 1)

let rec generate_pattern k =
  if k = 0 then [0]
  else begin
    let highest = highest_bit k in
    highest - 1 :: List.map (( lor ) highest) (generate_pattern (k land lnot highest))
  end

let () =
  let n, k = Scanf.scanf "%d %d\n" (fun n k -> n, k) in
  let abs = Array.init n (fun _ -> Scanf.scanf "%d %d\n" (fun a b -> a, b)) in
  generate_pattern k
  |> List.map (fun pat ->
      Array.fold_left (fun b' (a, b) ->
        if a land lnot pat = 0 then b + b' else b') 0 abs)
  |> List.fold_left max 0
  |> Printf.printf "%d\n"